
Example

Logic Program:

father(abraham, isaac) ←
mother(sarah, isaac) ←
father(isaac , jacob) ←

parent(X ,Y ) ← father(X ,Y )
parent(X ,Y ) ← mother(X ,Y )

ancestor(X ,Y ) ← parent(X ,Y )
ancestor(X ,Z ) ← parent(X ,Y ), ancestor(Y ,Z )

Query:

(∃X )(∃Y )ancestor(X ,Y )?

Answer:
Yes for X = abraham,Y = isaac; X = sarah,Y = isaac;
X = abraham,Y = jacob.

Programming with Prolog

Problem

Program

Query

Yes/No

Substitution(s)

Solution(s)

Modeling

Searching
proofs

Interpretation

First-Order Logic

T |= (∃X )(∃Y )ancestor(X ,Y ) iff

for all interpretations I holds
I |= T ⇒ I |= (∃X )(∃Y )ancestor(X ,Y ) iff

there does not exist an interpretation I such that I |= T but
I 6|= (∃X )(∃Y )ancestor(X ,Y ) iff

there does not exist an interpretation I such that I |= T and
I |= (∀X )(∀Y )¬ancestor(X ,Y ) iff

there does not exist an interpretation I such that
I |= T ∪ {(∀X )(∀Y )¬ancestor(X ,Y )} iff

T ∪ {(∀X )(∀Y )¬ancestor(X ,Y )} is unsatisfiable

First-Order Logic

(a) father(abraham, isaac)
(b) mother(sarah, isaac)
(c) father(isaac , jacob)
(d) ¬father(X ,Y ) ∨ parent(X ,Y )

(e) ¬mother(X ,Y ) ∨ parent(X ,Y )

(f) ¬parent(X ,Y ) ∨ ancestor(X ,Y )

(g) ¬parent(X ,Y ) ∨ ¬ancestor(Y ,Z ) ∨ ancestor(X ,Z )

(1) ¬ancestor(X ,Y ) (Query)
(2) ¬parent(X ,Y ) (Resolution of 1 and f using θ1 = {})
(3) ¬father(X ,Y ) (Resolution of 2 and d using θ2 = {})
(4) ⊥ (Resolution of 3 and a using θ3 = {X/abraham,Y /isaac})

Logic Programming

P |= (∃X )(∃Y )ancestor(X ,Y ) iff

P ∪ {(∀X )(∀Y )¬ancestor(X ,Y )} is unsatisfiable iff

P ∪ {← ancestor(X ,Y )} is unsatisfiable

Logic Programming

(a) father(abraham, isaac)←
(b) mother(sarah, isaac)←
(c) father(isaac , jacob)←
(d) parent(X ,Y )← father(X ,Y )

(e) parent(X ,Y )← mother(X ,Y )

(f) ancestor(X ,Y )← parent(X ,Y )

(g) ancestor(X ,Z )← parent(X ,Y ), ancestor(Y ,Z )

(1) ← ancestor(X ,Y ) (Query)
(2) ← parent(X ,Y ) (Resolution of 1 and f using θ1 = {})
(3) ← father(X ,Y ) (Resolution of 2 and d using θ2 = {})
(4) ← (Resolution of 3 and a using θ3 = {X/abraham,Y /isaac})

Resolution for Definite Logic Programs

SLD-resolution ≡ Linear resolution with Selection function for
Definite clauses.

Definition (Definite Goal)

A definite goal is a rule of the form

← A1, . . . ,An

where 0 ≤ n and each Ai , 0 < i ≤ n, is an atom.

Definition (Resolvent)

Let G be a definite goal ← A1, . . . ,Ak−1,Ak ,Ak+1, . . . ,Am,
Ak be a selected atom, and r be a definite rule B0 ← B1, . . . ,Bn.
We say that a goal G ′ is a resolvent derived from G and r using θ
if θ is the most general unifier of Ak and B0 and G ′ has the form
← (A1, . . . ,Ak−1,B1, . . . ,Bn,Ak+1, . . . ,Am)θ.

SLD-Derivation

Definition (SLD-derivation)

Let P be a definite logic program and G be a definite goal. An
SLD-derivation of P ∪ {G} is a (posibly infinite) sequence of goals
G = G0, . . . ,Gi , . . . , where each Gi+1 is a resolvent obtained from
Gi and a rule ri+1 from P using θi+1.

Definition (Successful, Failed, and Infinite Derivation)

A successful derivation ends in empty goal ←. A failed derivation
ends in non-empty goal with the property that all atoms does not
unify with the head of any rule. An infinite derivation is an infinite
sequence of goals.

SLD-Tree

Definition (SLD-Tree)

Let P be a definite logic program and G be a definite goal.
An SLD-tree for P ∪ {G} is a minimal tree satisfying the following:

Each node of the tree is a (possibly empty) definite goal
The root is G
If G ′ is a node of the tree and G ′′ is a resolvent derived from
G ′, then G ′ has a child G ′′

Standard Prolog
selects the first literal in the goal
chooses rules for unification in order as they appear in the
logic program
uses depth-first search strategy

Answer

Definition (Correct Answer)

Let P be a definite logic program and G be a definite goal
← A1, . . . ,An. An answer for P ∪{G} is a substitution for variables
in G . An answer θ for P ∪ {G} is correct iff P |= (A1, . . . ,An)θ.

Definition (Computed Answer)

Let G0, . . . ,Gn be a successful derivation using θ1, . . . , θn. Then
θ1 . . . θn restricted to the variables of G is the computed answer.

Soundness and Completeness

Theorem (Soundness)

Let P be a definite logic program and G be a definite goal. Every
computed answer for P ∪ {G} is a correct answer for P ∪ {G}.

Theorem (Completeness)

Let P be a definite logic program and G be a definite goal. For
every correct answer θ for P ∪ {G} there exists a computed answer
σ for P ∪ {G} and a substitution γ such that θ = σγ.

Fact (Termination)

SLD-resolution may not terminate.

SLDNF-Resolution

SLD-resolution augmented by the negation as failure rule.

Definition (Normal Goal)

A normal goal is a rule of the form

← L1, . . . , Ln

where 0 ≤ n and each Li , 0 < i ≤ n, is a literal.

Definition (Resolvent)

Let G be a normal goal ← L1, . . . , Lk−1, Lk , Lk+1, . . . , Lm,
Lk be a selected atom A, and r be a normal rule B0 ← M1, . . . ,Mn.
We say that a goal G ′ is a resolvent derived from G and r using θ
if θ is the most general unifier of Lk and B0 and G ′ has the form
← (L1, . . . , Lk−1,M1, . . . ,Mn, Lk+1, . . . , Lm)θ.

Negation as Failure

Definition (Negation as Failure Rule)

Let G be a normal goal ← L1, . . . , Lk−1, Lk , Lk+1, . . . , Lm and
Lk be a selected negated atom ∼A. We say that a normal goal G ′

is obtained from G using negation as failure rule if
P ∪ {← A} has finitely failed SLDNF-tree and G ′ has the form
← L1, . . . , Lk−1, Lk+1, . . . , Lm.

SLDNF-Derivation

Definition (SLDNF-Derivation)

Let P be a normal logic program and G be a normal goal.
An SLDNF-derivation of P ∪ {G} is a (possibly infinite) sequence
of goals G = G0, . . . ,Gi , . . . where each Gi+1

is derived from Gi and a rule ri+1 from P using θi+1, or
is obtained from Gi using negation as failure rule on selected
literal ∼A. In such case, ri+1 = ← A and θi+1 is identity.

Definition (Successful, Failed, and Infinite Derivation)

A successful derivation ends in empty goal ←. A failed derivation
ends in non-empty goal with the property that the selected literal is

an atom which do not unify with the head of any rule, or
a negated atom which do not have finitely failed SLDNF-tree.

An infinite derivation is an infinite sequence of goals.

SLDNF-Tree

Definition (SLDNF-Tree)

Let P be a normal logic program and G be a normal goal.
An SLDNF-tree for P ∪ {G} is a minimal tree satisfying the
following:

Each node of the tree is a (possibly empty) normal goal
The root is G
If G ′ is a node of the tree and G ′′ is a resolvent derived from
G ′, then G ′ has a child G ′′

If G ′ is a node of the tree and G ′′ is obtained from G ′ using
negation as failure rule, then G ′ has a child G ′′

Definition (Finitely Failed SLDNF-Tree)

A finitely failed SLDNF-tree is finite and has only failed branches.

SLDNF-Derivation and SLDNF-Tree

Please note, that SLDNF-tree is defined in terms of
SLDNF-derivation, and SLDNF-derivation is defined in terms of
SLDNF-tree. Such cyclic definitions are not correct. Proper
definitions are much more complex, although they capture the same
idea. They can be found in:

Lloyd, J. W. (1987). Foundations of Logic Programming. Springer.

Answer

Definition (Correct Answer)

Let P be a normal logic program and G be a normal goal
← L1, . . . , Ln. An answer for P ∪ {G} is a substitution for
variables in G . An answer θ for P ∪ {G} is correct iff
Comp(P) |= (L1, . . . , Ln)θ.

Definition (Computed Answer)

Let G0, . . . ,Gn be a successful derivation using θ1, . . . , θn. Then
θ1 . . . θn restricted to the variables of G is the computed answer.

Soundness and Completeness

Theorem (Soundness)

Let P be a normal logic program and G be a normal goal. Every
computed answer for P ∪ {G} is a correct answer for P ∪ {G}.

Fact (Termination)

SLDNF-resolution may not terminate.

Fact (Completeness)

SLDNF-resolution is not complete. Even if it terminates, it may not
compute all answers (see floundering).

Floundering

man(dilbert). man(bill).
husband(bill).
single(X) :- man(X), \+ husband(X).

?- single(X).
X = dilbert; No

man(dilbert). man(bill).
husband(bill).
single(X) :- \+ husband(X), man(X).

?- single(X).
No

Floundering

What is the nature of floundering problem?

If we want to resolve ← ∼ husband(X ), according to the “negation
as failure” rule, we check whether P ∪ {← husband(X )} has finitely
failed SLDNF-tree.
Recall that ← ∼ husband(X ) stands for ∼(∃X )∼ husband(X ),
and ← husband(X ) stands for ∼(∃X )husband(X ). They are not
complementary formulas!

On the other hand, if we want to resolve ← ∼ husband(dilbert),
we check whether P ∪ {← husband(dilbert)} has finitely failed
SLDNF-tree. In this case, ∼ husband(dilbert) and husband(dilbert)
are complementary. Flounering problem can occur only when we
resolve negated atom containing a variable.

Ordering of Rules Matters

on(a, b). on(b, c).
above(X, Y) :- on(X, Y).
above(X, Y) :- above(X, Z), on(Z, Y).

?- above(a, c).
Yes;
Error: Stack overflow.

on(a, b). on(b, c).
above(X, Y) :- above(X, Z), on(Z, Y).
above(X, Y) :- on(X, Y).

?- above(a, c).
Error: Stack overflow.

Martin Baláž, Martin Homola Lecture 8: Prolog



Ordering of Literals Matters

on(a, b). on(b, c).
above(X, Y) :- on(X, Y).
above(X, Y) :- above(X, Z), on(Z, Y).

?- above(a, c).
Yes;
Error: Stack overflow.

on(a, b). on(b, c).
above(X, Y) :- on(X, Y).
above(X, Y) :- on(Z, Y), above(X, Z).

?- above(a, c).
Yes;
No.

